

1. (Taha) Uma empresa usa 4 caminhões-tanques especiais para transportar 4 produtos de gasolina a seus clientes. Cada caminhão-tanque tem 4 compartimentos com diferentes capacidades: 750, 1.200, 1.500 e 1.750 litros específicos para carregar os produtos 1, 2, 3 e 4. As demandas diárias dos 4 produtos são de 10, 15, 12 e 8 mil litros. Quaisquer quantidades que não possam ser transportada pela frota da empresa devem ser transportadas por uma subcontratada, a um custo adicional de 5, 12, 8 e 10 centavos por litro para os produtos 1, 2, 3 e 4 respectivamente. Desenvolva um modelo para determinar a distribuição dos produtos com o menor custo adicional de subcontratação.

1.1 (Adaptação 1) Uma empresa usa 1 caminhão-tanque especial para transportar 2 produtos de gasolina a seus clientes. Cada caminhão-tanque tem 3 compartimentos com diferentes capacidades: 7, 5 e 6 ml litros para carregar os produtos. As demandas diárias dos 2 produtos são de 6 e 5 mil litros. O custo para transportar os combustíveis depende do tanque onde ele é armazenado, sendo de \$50, \$70 e \$40 para cada mil litros de qualquer combustível carregado nos tanques 1, 2 e 3 respectivamente. Implemente um modelo Gusek para distribuir os produtos com o menor custo possível.

```

set TQ := {1..3};    # conjunto de tanques
set PD := {1..2};    # conjunto de produtos

param custo{t in TQ};    # preço de transporte em cada tanque (mil litros)
param cap{TQ};          # capacidade de cada tanque
param dem{PD};          # demanda de cada produto
var x{PD, TQ}, >= 0;    # quantidade de litros (em milhar) do produto i no tanque j
var y{PD, TQ}, binary; # 1 se o produto p usar o tanque t e 0 cc.

minimize Z: sum{i in PD, j in TQ} x[i, j] * custo[j];
demanda{p in PD}: sum{t in TQ} x[p, t] >= dem[p];
capacidade{t in TQ, p in PD}: x[p, t] <= cap[t]*y[p, t];
unicidade{t in TQ}: sum{p in PD} y[p, t] <= 1;

solve;
data;
param custo :=
  1 50
  2 70
  3 40;
param cap :=
  1 7
  2 5
  3 6;
  param dem:=
  1 6
  2 5;
end;

```

1.1 (Adaptação 1) Uma empresa usa 2 caminhão-tanque especial para transportar 2 produtos de gasolina a seus clientes. Cada caminhão-tanque tem 3 compartimentos com diferentes capacidades: 7, 5 e 6 ml litros para carregar os produtos. As demandas diárias dos 2 produtos são de 12 e 18 mil litros. O custo para transportar os combustíveis depende do tanque onde ele é armazenado, sendo de \$50, \$70 e \$40 para cada mil litros de qualquer combustível carregado nos tanques 1, 2 e 3 respectivamente. Implemente um modelo Gusek para distribuir os produtos com o menor custo possível.

```

set TQ := {1..3};    # conjunto de tanques
set PD := {1..2};    # conjunto de produtos
set CT := {1..2};    # conjunto de caminhões

param custo{t in TQ};    # preço de transporte em cada tanque (mil litros)
param cap{TQ};           # capacidade de cada tanque
param dem{PD};           # demanda de cada produto
var x{PD, TQ, CT}, >= 0;    # quantidade de litros (em milhar) do produto i no tanque j
var y{PD, TQ, CT}, binary; # 1 se o produto p usar o tanque t e 0 cc.

minimize Z: sum{i in PD, j in TQ, c in CT} x[i, j, c] * custo[j];
demanda{p in PD}: sum{t in TQ, c in CT} x[p, t, c] >= dem[p];
capacidade{p in PD, t in TQ, c in CT}: x[p, t, c] <= cap[t]*y[p, t, c];
unicidade{c in CT, t in TQ}: sum{p in PD} y[p, t, c] <= 1;

solve;
data;
param custo :=
  1 50
  2 70
  3 40;
param cap :=
  1 7
  2 5
  3 6;
  param dem:=
  1 12
  2 18;
end;

```